var rad:Number=250;
var posX:Number=stage.stageWidth/2;
var posY:Number=stage.stageHeight/2;
var thumbWidth:Number=70;
var thumbHeight:Number=53;
var thumbsArray:Array=[];
var picsArray:Array=[];
var holdersArray:Array=[];
var midsArray:Array=[];
var jLen:Vector.<Number>=new Vector.<Number>();
jLen=Vector.<Number>([1,6,10,12,10,6,1]);
var thetaStep:Vector.<Number>=new Vector.<Number>();
thetaStep=Vector.<Number>([0,60,36,30,36,60,0]);
var phiStep:Number=30;
var phiTilt:Vector.<Number>=new Vector.<Number>();
phiTilt=Vector.<Number>([-90,-60,-30,0,30,60,90]);
var autoOn:Boolean=true;
var manualOn:Boolean=false;
var prevX:Number;
var prevY:Number;
this.transform.perspectiveProjection.fieldOfView=70;
var board:Sprite=new Sprite();
this.addChild(board);
board.x=posX;
board.y=posY;
drawBoard();
box1.mouseEnabled=false;
box1.wordWrap=true;
box1.text="Press and move the mouse over the sphere to rotate. DOUBLE CLICK a thumbnail to load an image.";
box2.mouseEnabled=false;
box2.wordWrap=true;
box2.text="";
box2.visible=false;
var photoHolder:Sprite=new Sprite();
this.addChild(photoHolder);
photoHolder.x=120;
photoHolder.y=160;
var loader:Loader=new Loader();
photoHolder.addChild(loader);
photoHolder.visible=false;
var spSphere:Sprite=new Sprite();
board.addChild(spSphere);
spSphere.x=0;
spSphere.y=0;
spSphere.z=rad;
setUpPics();
buildSphere();
spSphere.rotationY=0;
spSphere.rotationX=0;
spSphere.rotationZ=0;
spSphere.filters=[new GlowFilter(0xfcfdf3,2.5,2.5,2.5)];
rotateSphere(0,0,0);
setUpListeners();
function drawBoard():void {
	  board.graphics.clear();
	  board.graphics.lineStyle(1,0x333333);
	  board.graphics.beginFill(0x000000);
	board.graphics.drawRect(-240,-240,480,480);
  board.graphics.endFill();
	  }
function setUpListeners():void {
	       var i:int;
	      var j:int;
	      this.addEventListener(Event.ENTER_FRAME,autoRotate);
	     board.addEventListener(MouseEvent.ROLL_OUT,boardOut);
                    board.addEventListener(MouseEvent.MOUSE_MOVE,boardMove);
                    board.addEventListener(MouseEvent.MOUSE_DOWN,boardDown);
	     board.addEventListener(MouseEvent.MOUSE_UP,boardUp);
	    loader.contentLoaderInfo.addEventListener(Event.COMPLETE,doneLoad);
	    loader.contentLoaderInfo.addEventListener(IOErrorEvent.IO_ERROR,loadingError);
	    photoHolder.addEventListener(MouseEvent.CLICK,holderClicked); 
		 for(i=0;i<7;i++){
		    for(j=0;j<jLen[i];j++){
				holdersArray[i][j].doubleClickEnabled=true;
			    holdersArray[i][j].addEventListener(MouseEvent.DOUBLE_CLICK,picClicked);
		       }
		  } 
	}
function holderClicked(e:MouseEvent):void {
	board.visible=true;
	photoHolder.visible=false;
	box1.text="Press and drag the mouse over the sphere to rotate. DOUBLE CLICK a thumbnail to see the full size image.";	
	manualOn=false;
	autoOn=true;
}
function picClicked(e:MouseEvent):void {
	var targName:String="";
	var i:int;
	var j:int;
	targName=e.currentTarget.name;
	i=int(targName.charAt(3));
	j=int(targName.substring(5,targName.length));
	board.visible=false;
	loader.load(new URLRequest(picsArray[i][j]));
	box1.text="";
	box2.text="Loading...";
	 box2.visible=true;
}
function loadingError(e:IOErrorEvent):void {
	box2.text="There has been an error loading the image. The server may be busy. Refresh the page and try again.";
}
function doneLoad(e:Event):void {
	box1.text="Click on image to close it.";
	photoHolder.visible=true;
	box2.text="";
    box2.visible=false;
  }
  function autoRotate(e:Event):void {
    if(autoOn && !manualOn){
			 spSphere.transform.matrix3D.prependRotation(-0.5,Vector3D.Y_AXIS); 
			 zSortPics();
		  } 
	} 
 function boardOut(e:MouseEvent):void {
			autoOn=true;
			manualOn=false;
	}
function boardDown(e:MouseEvent):void {	
	        prevX=board.mouseX;
			prevY=board.mouseY;
			autoOn=false;
			manualOn=true;
	}
function boardUp(e:MouseEvent):void {
			manualOn=false;
	}
     function boardMove(e:MouseEvent):void { 
		        var locX:Number=prevX;
				var locY:Number=prevY;
				if(!autoOn && manualOn){
				prevX=board.mouseX;
				prevY=board.mouseY;
				rotateSphere(prevY-locY,-(prevX-locX),0);
				e.updateAfterEvent();
				}
	}
function setUpPics():void {
	thumbsArray[0]=[new Bitmap(new small1(70,53))]; 
	picsArray[0]=["spherical_gallery/pic1.jpg"];
	thumbsArray[1]=[new Bitmap(new small2(70,53)),new Bitmap(new small3(70,53)),new Bitmap(new small4(70,53)),new Bitmap(new small5(70,53)),new Bitmap(new small6(70,53)),new Bitmap(new small7(70,53))];
	picsArray[1]=["pic2.jpg","pic3.jpg","pic4.jpg","pic5.jpg","pic6.jpg","pic7.jpg"];
	thumbsArray[2]=[new Bitmap(new small8(70,53)),new Bitmap(new small9(70,53)),new Bitmap(new small10(70,53)),new Bitmap(new small11(70,53)),new Bitmap(new small12(70,53)),new Bitmap(new small13(70,53)),new Bitmap(new small14(70,53)),new Bitmap(new small15(70,53)),new Bitmap(new small16(70,53)),new Bitmap(new small17(70,53))];
	picsArray[2]=["pic8.jpg","pic9.jpg","pic10.jpg","pic11.jpg","pic12.jpg","pic13.jpg","pic14.jpg","pic15.jpg","pic16.jpg","pic17.jpg"];
	thumbsArray[3]=[new Bitmap(new small18(70,53)),new Bitmap(new small19(70,53)),new Bitmap(new small20(70,53)),new Bitmap(new small21(70,53)),new Bitmap(new small22(70,53)),new Bitmap(new small23(70,53)),new Bitmap(new small24(70,53)),new Bitmap(new small25(70,53)),new Bitmap(new small26(70,53)),new Bitmap(new small27(70,53)),new Bitmap(new small28(70,53)),new Bitmap(new small29(70,53))];
	picsArray[3]=["pic18.jpg","pic19.jpg","pic20.jpg","pic21.jpg","pic22.jpg","pic23.jpg","pic24.jpg","pic25.jpg","pic26.jpg","pic27.jpg","pic28.jpg","pic29.jpg"];
	thumbsArray[4]=[new Bitmap(new small30(70,53)),new Bitmap(new small31(70,53)),new Bitmap(new small32(70,53)),new Bitmap(new small33(70,53)),new Bitmap(new small34(70,53)),new Bitmap(new small35(70,53)),new Bitmap(new small36(70,53)),new Bitmap(new small37(70,53)),new Bitmap(new small38(70,53)),new Bitmap(new small39(70,53))];
	picsArray[4]=["pic30.jpg","pic31.jpg","pic32.jpg","pic33.jpg","pic34.jpg","pic35.jpg","pic36.jpg","pic37.jpg","pic38.jpg","pic39.jpg"];
	thumbsArray[5]=[new Bitmap(new small40(70,53)),new Bitmap(new small41(70,53)),new Bitmap(new small42(70,53)),new Bitmap(new small43(70,53)),new Bitmap(new small44(70,53)),new Bitmap(new small45(70,53))];
	
	picsArray[5]=["pic40.jpg","pic41.jpg","pic42.jpg","pic43.jpg","pic44.jpg","pic45.jpg"];
	thumbsArray[6]=[new Bitmap(new small46(70,53))];
	picsArray[6]=["pic46.jpg"];
}
function buildSphere():void {
	var i:int;
	var j:int;
	var tStep:Number;
	var pStep:Number=phiStep*Math.PI/180;
	for(i=0;i<7;i++){
		holdersArray[i]=[];
		midsArray[i]=[];
		tStep=thetaStep[i]*Math.PI/180;
		for(j=0;j<jLen[i];j++){
			midsArray[i][j]=new Vector3D(rad*Math.sin(i*pStep)*Math.sin(j*tStep),-rad*Math.cos(i*pStep),-rad*Math.sin(i*pStep)*Math.cos(j*tStep));
			holdersArray[i][j]=new Sprite();
		    holdersArray[i][j].name="pic"+String(i)+"_"+String(j);
			holdersArray[i][j].addChild(thumbsArray[i][j]);
			thumbsArray[i][j].x=-thumbWidth/2;
			thumbsArray[i][j].y=-thumbHeight/2;
			spSphere.addChild(holdersArray[i][j]);
			holdersArray[i][j].x=midsArray[i][j].x;
			holdersArray[i][j].y=midsArray[i][j].y;
			holdersArray[i][j].z=midsArray[i][j].z;
			
			holdersArray[i][j].rotationX=phiTilt[i];
			holdersArray[i][j].rotationY=-j*thetaStep[i];
			}
	}
	  zSortPics();
}
function zSortPics():void {
	var distArray:Array=[];
	var dist:Number;
	var i:int;
	var j:int;
	var k:int;
	var curMatrix:Matrix3D;
	var curMid:Vector3D;
	curMatrix=spSphere.transform.matrix3D.clone();
	while(spSphere.numChildren>0){
		spSphere.removeChildAt(0);
		}
		for(i=0;i<7;i++){
			for(j=0;j<jLen[i];j++){
			curMid=curMatrix.deltaTransformVector(midsArray[i][j]);
		    dist=curMid.z;
		    distArray.push([dist,i,j]);
			}
		
		}
	  distArray.sort(byDist); 
	  for(k=0;k<distArray.length;k++){
		  spSphere.addChild(holdersArray[distArray[k][1]][distArray[k][2]]); 
		  holdersArray[distArray[k][1]][distArray[k][2]].alpha=Math.max(k/(distArray.length-1),0.5);		}
}
function byDist(v:Array,w:Array):Number {
	 if (v[0]>w[0]){
		return -1;
	  } else if (v[0]<w[0]){
		return 1;
	   } else {
		return 0;
	  } 
   }
function rotateSphere(rotx:Number,roty:Number,rotz:Number):void {
	  spSphere.z=0;
	  spSphere.transform.matrix3D.appendRotation(rotx,Vector3D.X_AXIS);
	  spSphere.transform.matrix3D.appendRotation(roty,Vector3D.Y_AXIS);
	  spSphere.transform.matrix3D.appendRotation(rotz,Vector3D.Z_AXIS); 
	  spSphere.z=rad; 
	  zSortPics();
}




